LED Matrix

Flashing Lights

The Matrix

- An LED Matrix is an LED-based display commonly used to display information such as time and scrolling text. The LEDs are located at every row and column intersection of the matrix, which can be seen in the picture to the left. Every grey circle in the picture represents an LED. The schematic also shows you the location of the LEDs inside the matrix.
- Helpful Explanation (1:30 to 6:45): https://youtu.be/G4llo-MRSiY?t=90





Wiring the Matrix

- VCC = 5V pin / rail
- GND = GND pin / Rail
- DIN = any Arduino Pin (2-13)
- CLK = any Arduino Pin (2-13)
- CS = any Arduino Pin (2-13)
- The breadboard can also be used



Library

The MaxMatrix library is used for easy control of the LED matrix.

Resource for installing the MaxMatrix library: https://electronoobs.com/eng_arduino_max_matrix.php#google_vignette

Constructor

Constructor:

MaxMatrix (DIN pin, CS pin, CLK pin, # of MAX7219 modules in use)

Parameter 1: Arduino pin that the DIN pin of LED Matrix is connected to

Parameter 2: Arduino pin that the CS pin of LED Matrix is connected to

Parameter 3: Arduino pin that the CLK pin of LED Matrix is connected to

Parameter 4: number of LED Matrixes (MAX7219) connected together

If DIN is in pin 12, CLK is in pin 11, and CS is in pin 10, and I was only using one LED matrix module, My constructor would be:

```
MaxMatrix my_matrix = MaxMatrix(12, 10, 11, 1);
```

Functions

- There are many functions for the LED matrix.
- init(): Initialize the object
- setIntensity(num): Set LED brightness (0-15)
- setDot(x,y,true): Turn led at (x,y) coordinate on
- clear(): Clear LEDs, turns all LEDs off
- writeSprite(x,y,sprite): Displays sprite (shape) at (x,y) coordinate
    - A sprite is a pointer to a byte or character array containing the data for a shape
- shiftLeft(rotate,fill_zero): shifts the shape to the left by one column
    - if rotate is true, the column that shifts out of the display will shift back in from the opposite side, creating a continuous loop
    - if fill_zero is true, as the shape moves through the display, the LEDs behind the shape will be set to 0 (off)
- shiftRight(rotate, fill_zero): shifts the shape to the right by one column
- shiftUp(rotate): shifts the shape up by one row
- shiftDown(rotate): shifts the shape down by one row

Arrays

- In programming, an array is a series of memory locations that are represented as a single item. They are used to create collections of items with similar datatypes.
- For example, an integer array holds a collection of integers. And a character array holds a collection of characters.
- Here are two examples of how to create an integer array of three integers.

```
int my_int_array[3] = {1,2,3};

// alternatively:

int my_int_array[3] = {
    1,
    2,
    3
};
```

Making an Array for a Sprite

- To make an array for a sprite, you must create a character array. (represented as char in C++).
- The first two items are the dimensions of the Matrix. (8x8) in this case.
- Then, there are 8 items that represent the sprite. Each item is a B(for binary) followed by a series of 0s and 1s that represent an activated or deactivated LED.
- For Example, B10100001 means that LEDs 1, 3, and 8 are on in the current row.
- If the displayed sprite appears rotated, the B values can be changed to be oriented correctly.
- To display the array, use writeSprite(x,y,sprite) in void loop like this:

```
my_matrix.writeSprite(0, 0, smiley_face);
```

```
void setup() {
 my_matrix.init();  // initialize the Matrix
 my_matrix.setIntensity(8);  // set brightness to
0 (range is 0-15)
 my_matrix.clear();  // clear the display (turn
off all LEDs)
}
```

```
char smiley_face[] = {
  8, // x dimension
  8,          // y dimension
  B11111111,
  B10000001,
  B10100101,
  B10000001,
  B10100101,
  B10111101,
  B10000001,
  B11111111
};
```



Task 8: Faces

Make character arrays for neutral and sad faces

LED Matrix: Scrolling Sprites

- It is possible for a sprite to move across the matrix.
- To do this, the sprite must be displayed, then cleared, then displayed in a different location. This process must repeat.
- To display a moving sprite of Pac-Man, start by displaying it on coordinated outside of the matrix, then using a for loop, move it one pixel to the right for every iteration.

```
char Pacman[] = {8, 8,
                 B00011100,
                 B00100010,
                 B01000001,
                 B01001001,
                 B01010101,
                 B00100010,
                 B00000000,
                 B00001000};

void setup() {
 my_matrix.init();  // initialize the Matrix
 my_matrix.setIntensity(8);  // set brightness to 0 (range is 0-15)
 my_matrix.clear();  // clear the display (turn off all LEDs)
}

void loop() {
 // start outside the matrix (to the left)
 my_matrix.writeSprite(-8,0,Pacman);

 /* move the Pacman 16 times to the right
  * first 8 times = Pacman is shifted in from the left
  * last 8 times = Pacman is shifted out to the right
  */
 for (int i=1; i<17; i++){
   // update the location of the Pacman
   my_matrix.clear();
   my_matrix.writeSprite(-8+i,0,Pacman);
   delay(100);
 }
 my_matrix.clear();  // clear the display
}
```

Scrolling Sprites

- Alternatively, the MD_MAX72XX and MD_Parola libraries can be used to scroll sprites across the LED matrix.
- To install the libraries:
    1. Sketch → Include Library → Manage Libraries
    2. This will open Arduino's Library Manager, where you can search for each library and install it

Once you have installed the libraries, you will have to "include" them in your Arduino sketch. There are two ways to do this:

1. Sketch → Include Library → "MD_Parola" & "MD_MAX72XX"
2. Use the "#include" keyword: #include <MD_Parola.h> and #include <MD_MAX72xx.h>

MD_Parola Library Constructor

MD_Parola (module type, DIN pin, CLK pin, CS pin, # of MAX7219 modules)

Parameter 1: type of MAX7219 module; ours is "GENERIC_HW"
Parameter 2: Arduino pin that the DIN pin of LED Matrix is connected to
Parameter 3: Arduino pin that the CLK pin of LED Matrix is connected to
Parameter 4: Arduino pin that the CS pin of LED Matrix is connected to
Parameter 5: number of LED Matrixes (MAX7219) connected together

- In the IDE, the constructor would look something like this:

- `MD_Parola my_matrix = MD_Parola(MD_MAX72XX::GENERIC_HW, 12, 11, 10, 1);`

MD_Parola Library Functions

- begin(): initialize the object
- setIntensity(num): set the intensity (brightness) of the display (range 0-15)
- displayClear(): clear the display (turn off all LEDs)
- displayScroll(text, text alignment, effect, speed): scrolling text display
  - text is the text you want to display on the LED Matrix
  - text alignment is how you want to align the text (left, center, right)
  - effect is the type of effect you want to use for the entry and exit of text on the display
  - speed is the time, in milliseconds, between animation frames

Function resource: https://majicdesigns.github.io/MD_Parola/class_m_d___parola.html

Enumerator	Text alignment options
PA_LEFT	The leftmost column for the first character will be on the left side of the display.
PA_CENTER	The text will be placed with equal number of blank display columns either side.
PA_RIGHT	The rightmost column of the last character will be on the right side of the display.

Enumerator	Effect options
PA_NO_EFFECT	Used as a place filler, executes no operation.
PA_PRINT	Text just appears (printed)
PA_SCROLL_UP	Text scrolls up through the display.
PA_SCROLL_DOWN	Text scrolls down through the display.
PA_SCROLL_LEFT	Text scrolls right to left on the display.
PA_SCROLL_RIGHT	Text scrolls left to right on the display.
PA_SPRITE	Text enters and exits using user defined sprite.

Code Breakdown

```
#include <MD_Parola.h>
#include <MD_MAX72xx.h>

MD_Parola my_matrix = MD_Parola(MD_MAX72XX::GENERIC_HW, 12, 11, 10, 1);

const char* text = {“Hello"};

void setup() {
  my_matrix.begin();
  my_matrix.setIntensity(5);
  my_matrix.displayClear();
  my_matrix.displayScroll("Hello", PA_CENTER, PA_SCROLL_LEFT, 100);
}

void loop() {

  if(my_matrix.displayAnimate()){
    my_matrix.displayReset();
  }

}
```

Include Libraries

Constructor

Initialize matrix

Set matrix brightness to 5 and clear display (all LEDs off)

This is an if, then statement. It follows the format:
If(condition){
Do this
}
If the condition is true, then whatever is inside of it will be executed by the computer. In this case, this statement perpetually resets the matrix, constantly allowing it to use displayScroll()

Display “Hello”, centered in the center of the Matrix, scrolling left, with 100ms of delay each time the display refreshes

```
const char* text = {“Hello"};
```

Pointers

- “text” is a pointer.
- In computer science, a pointer is a variable that stores the memory address of another variable or a value. In other words, it stores where a value is stored, instead of storing the value itself.
- In this case, text acts as a string, which is a collection of characters.

Task 9: Scroll

- Use the LED matrix to output a smiley face symbol :) and use PA_SCROLL to animate it. Experiment with the arguments of the displayScroll() function.

displayScroll(text, text alignment, effect, speed)

More effect options

Enumerator	
PA_NO_EFFECT	Used as a place filler, executes no operation.
PA_PRINT	Text just appears (printed)
PA_SCROLL_UP	Text scrolls up through the display.
PA_SCROLL_DOWN	Text scrolls down through the display.
PA_SCROLL_LEFT	Text scrolls right to left on the display.
PA_SCROLL_RIGHT	Text scrolls left to right on the display.
PA_SPRITE	Text enters and exits using user defined sprite.
PA_SLICE	Text enters and exits a slice (column) at a time from the right.
PA_MESH	Text enters and exits in columns moving in alternate direction (U/D)
PA_FADE	Text enters and exits by fading from/to 0 and intensity setting.
PA_DISSOLVE	Text dissolves from one display to another.
PA_BLINDS	Text is replaced behind vertical blinds.
PA_RANDOM	Text enters and exits as random dots.
PA_WIPE	Text appears/disappears one column at a time, looks like it is wiped on and off.
PA_WIPE_CURSOR	WIPE with a light bar ahead of the change.
PA_SCAN_HORIZ	Scan the LED column one at a time then appears/disappear at end.
PA_SCAN_HORIZX	Scan a blank column through the text one column at a time then appears/disappear at end.
PA_SCAN_VERT	Scan the LED row one at a time then appears/disappear at end.
PA_SCAN_VERTX	Scan a blank row through the text one row at a time then appears/disappear at end.
PA_OPENING	Appear and disappear from the center of the display, towards the ends.
PA_OPENING_CURSOR	OPENING with light bars ahead of the change.
PA_CLOSING	Appear and disappear from the ends of the display, towards the middle.
PA_CLOSING_CURSOR	CLOSING with light bars ahead of the change.
PA_SCROLL_UP_LEFT	Text moves in/out in a diagonal path up and left (North East)
PA_SCROLL_UP_RIGHT	Text moves in/out in a diagonal path up and right (North West)
PA_SCROLL_DOWN_LEFT	Text moves in/out in a diagonal path down and left (South East)
PA_SCROLL_DOWN_RIGHT	Text moves in/out in a diagonal path down and right (North West)
PA_GROW_UP	Text grows from the bottom up and shrinks from the top down.
PA_GROW_DOWN	Text grows from the top down and and shrinks from the bottom up.

ULTRA**SONIC** SENSOR

Sound bounces

WHAT IS AN ULTRASONIC SENSOR?

- An ultrasonic sensor is a sensor that uses sound to measure distance from it to an object.
- The sensor consists of a transmitter and a receiver. The transmitter emits ultrasonic sound waves beyond human hearing range, and the receiver catches these waves.
- Distance is determined based on how long the waves take to make a journey from the transmitter to the receiver after bouncing off an object. The farther the object is, the longer it takes for the ultrasonic waves to travel.
- Based on the time, the distance can be computed using this formula: distance = (duration / 2) * 0.034



THE DISTANCE EQUATION

- (duration / 2) * 0.034 is used to calculate the distance. How does it work?
- Duration is divided by 2 because it is the duration for the 2-way trip.
- The speed of sound is 340 m/s. However, the duration measured by the ultrasonic sensor is in microseconds (μs), and we want the distance in centimeters. Therefore, we must convert the speed of sound to cm/μs, which is 0.034 cm/μs. Now, following the formula: distance = speed x time, we derive the following equation: distance = (duration / 2) * 0.034

WIRING – SENSOR & TRAFFIC LIGHT



The trig, echo, and all the traffic light data pins are connected to the Arduino. Vcc is connected to a power rail, and GND is connected to a ground rail

EXAMPLE: PROXIMITY LIGHTS

- What if I want to write code using the sensor where different LEDs light up depending on the distance of a target?
- If the target is within 50 cm of the sensor, the green light would turn on.
- If the target is between 51 and 75 cm of the sensor, the yellow light would turn on.
- If the target is more than 75 cm away from the sensor, the red light would turn on.

```
const int trig = 2;
const int echo = 3;

long duration;
int distance;

const int red = 8;
const int yellow = 9;
const int green =  10;

void setup() {

  pinMode(red, OUTPUT);
  pinMode(yellow, OUTPUT);
  pinMode(green, OUTPUT);

  pinMode(trig, OUTPUT);
  pinMode(echo, INPUT);

  Serial.begin(9600);

}

void loop() {

  digitalWrite(trig, LOW);
  delayMicroseconds(2);

  digitalWrite(trig, HIGH);
  delayMicroseconds(10);
  digitalWrite(trig, LOW);

  duration = pulseIn(echo, HIGH);

  distance = (duration/2) * 0.034;


  if (distance < 50){
    digitalWrite(green, HIGH);
    digitalWrite(yellow, LOW);
    digitalWrite(red, LOW);
  }

  else if (distance > 50 && distance < 75){
    digitalWrite(green, LOW);
    digitalWrite(yellow, HIGH);
    digitalWrite(red, LOW);
  }

  else{
    digitalWrite(green, LOW);
    digitalWrite(yellow, LOW);
    digitalWrite(red, HIGH);
  }

  Serial.println(distance);

}
```

CODE

CODE - VARIABLES

```
const int trig = 2;
const int echo = 3;

long duration;
int distance;

const int red = 8;
const int yellow = 9;
const int green =  10;
```

- The variables for the trig, echo, and LED pins are all const int, because they are integers that are constant, and do not need to be changed.
- Duration is a long, which is a signed 32-bit number. They can range from -2,147,483,648 to 2,147,483,647
- Distance is int rather than const int because it is not constant. It is frequently measured and changed.

CODE - PINMODE

```
void setup() {

  pinMode(red, OUTPUT);
  pinMode(yellow, OUTPUT);
  pinMode(green, OUTPUT);

  pinMode(trig, OUTPUT);
  pinMode(echo, INPUT);

  Serial.begin(9600);

}
```

- All of the pins are outputs except for the echo pin, which is an input.
- The trig (trigger) pin is used to trigger an ultrasonic sound pulse, while echo receives it after it bounces off of an object.
- Since echo must measure this pulse and tell the Arduino microcontroller about it, it needs an input pin.

```
void loop() {

  digitalWrite(trig, LOW);
  delayMicroseconds(2);

  digitalWrite(trig, HIGH);
  delayMicroseconds(10);
  digitalWrite(trig, LOW);

  duration = pulseIn(echo, HIGH);

  distance = (duration/2) * 0.034;
```

CODE - PULSE

- Trig, which triggers an ultrasonic pulse is disabled for 2 microseconds (for reference, there are 1000 microseconds in a millisecond).
- Then, trig is activated for 10 microseconds, sending out an ultrasonic pulse.
- The time that the pulse takes to reach the object then back to the Arduino is recorded and moved to the variable *duration* using the pulseIn function.
- The distance formula is used to calculate the distance based off of the duration.

```
  if (distance < 50){
    digitalWrite(green, HIGH);
    digitalWrite(yellow, LOW);
    digitalWrite(red, LOW);
  }

  else if (distance > 50 && distance < 75){
    digitalWrite(green, LOW);
    digitalWrite(yellow, HIGH);
    digitalWrite(red, LOW);
  }

  else{
    digitalWrite(green, LOW);
    digitalWrite(yellow, LOW);
    digitalWrite(red, HIGH);
  }

  Serial.println(distance);

}
```

CODE – IF, ELSE IF, ELSE

- If the distance is less than 50, all three of the LEDs are turned off except for the green one.
- ***else if*** is generally used in situations where more than one ***if*** statement is needed. The ***else if*** statement will only be checked if the ***if*** statement is false.
- This is different from just ***if*** statements because with ***if*** statements, they are all checked, whether or not any are false.
- The ***else*** statement is like an ***else if*** statement with no condition. If all of the ***if*** and ***else if*** statements before it are false, whatever is in the ***else*** block will run.
- Serial.println(distance) is used to print the distance to the serial monitor. This is not necessary, but it can be used to keep track of the distance and diagnose potential problems with the setup/code.

TASK 10: ULTRASONIC BEEPS

- Use a piezo buzzer and an ultrasonic sensor to create a proximity siren.
- The buzzer should play a repeated beeping sound (like the beep boop exercise).
- It should play a higher-pitched tone with more rapid beeps as the target comes closer to the sensor.



TASK 11:LOUD SPIRALS

- Attach a NeoPixel ring, an ultrasonic sensor, and a buzzer to the Arduino.
- Modify your code from task 10 to include the NeoPixel ring.
- Make the ring create a spiral of different colours depending on the distance between the sensor and the target.
- 0-50 cm: red spiral
- 50-100 cm: yellow spiral
- 100+ cm: green spiral



Analog and Digital

Analog and digital are words that are commonly used to differentiate between items that share a purpose.

A common example are clocks. Analog clocks use hands to display the time while digital clocks use a digital display.

Another example are analog and digital music formats. Examples of digital audio formats are mp3 and FLAC files, while examples of analog ones are vinyl records and cassette tapes.

What is the Difference?

- Analog information can be read as electric pulses of varying amplitude, while digital information can only be binary, read as only one of two possible amplitudes.

Analog Circuits vs Digital Circuits



Example of Analog Signal



Example of Digital Signal

Arduino Analog pins

- The Arduino Uno has 5 analog pins, labeled A0 to A5.
- These pins can be used to input and output analog signals.
- The Arduino has a built-in analog to digital converter that converts analog signals to digital and back for the Arduino to use.
- Using an analog pin is very similar to using a digital pin. The functions simply use *analog* instead of *digital*.
- Analog functions would not use HIGH or LOW, but rather a numerical value between 0 and 1023. This number is limited by the Arduino's 10 bit analog-to-digital converter.
- For example, digitalWrite(3, HIGH); could be analogWrite(A3, 1023);



Light **D**ependent **R**esistor

Or LDR,
or photoresistor



What is an LDR?

- An LDR is a resistor that has a variable resistance based on the intensity of the light that it is exposed to.
- For example, if the LDR is in direct sunlight, the resistance will be lower than if it were in a dark room.
- In the LDR is a cadmium sulphide track. When light hits it, the photons of the light excite the valence electrons in the LDR, which reduces the material's conductivity, which then increases the resistance.





Real World Applications

- LDR’s can be used for many applications:
  - Automatic streetlights
  - Automatic night lights
  - Mobile phone auto-brightness
  - Automatic camera flash control

LDR → Arduino

- Connect the LDR in series to an **analog** Arduino pin (labeled A0-A5).
- Include a 550Ω resistor.





LDR

- The LDR value is essentially the voltage drop across the LDR. This gives us an indication of the LDR's resistance, since according to Ohm's Law, current stays the same in a series circuit. Therefore, the voltage drop across a load depends on its resistance. We need to use an additional resistor in our circuit because if the LDR was the only load in the circuit, all 5 volts will be supplied to the LDR, leading to a voltage drop of 5 volts.



Programming the LDR

- The LDR uses the analogRead() function to read a value from a component.
- Because of this, pinMode(*pin #*, INPUT) will be used.
- The value that the Arduino reads from the LDR can be printed to the serial monitor.

Example: Print to Serial Monitor

```
const int LDR_pin = 0;
int LDR_value = 0;

void setup() {
  Serial.begin(9600);
}

void loop() {
  LDR_value = analogRead(LDR_pin);
  Serial.print("LDR Value is: ");
  Serial.println(LDR_value);
  delay(200);
}
```

LDR_pin is a const int because it is the LDR pin and doesn't change.

LDR_value is not a const int because the value changes frequently.

The LDR_value is the value that is read from the LDR_pin by the Arduino. It is then printed to the serial monitor. It is refreshed every 200 milliseconds.

Task 12: Night Light

- Use the LDR and the LED matrix to create an automatic night light.
- When the room is dark, Every pixel on the matrix should light up.



Task 13: Adaptive Night Light

- Use the LDR and the LED matrix to create an automatic night light that changes the brightness of the LED matrix based on the brightness of the room.
- Include three different brightness modes.

I2C LCD

Readable text

WHAT IS AN LCD?

LCD stands for **L**iquid **C**rystal **D**isplay.

In an LCD, there is a layer of liquid crystal material that are sandwiched between electrodes. On either side of this is a polarizer that only lets certain light waves out.

The liquid crystals bend the light for it to pass through the polarizers.

Colour LCDs In front of this system is an RGB filter that is used to display either of the three colours.

There is also a backlight that illuminates the display. Otherwise, it would be too dark to see.

More: https://www.youtube.com/watch?v=VamqtyatBss&ab_channel=AKIOTV



16 x 2 LCD

- This LCD has a 16x2 grid of spaces, with each space containing a smaller 5x8 grid that can be used for displaying characters and sprites.
- It is used with an I2C for easy operation.





WHAT IS I2C?

- **I2C** stands for **I**nter-**I**ntegrated **C**ircuit.
- It is used for easier communication between the Arduino and the LCD.
- It utilizes the serial data (SDA) and serial clock (SCL) pins on the Arduino microcontroller.

LCD + I2C



- An I2C bus can be attached to a 16x2 LCD for easier control of it. The module can then be connected to the SDA and SCL pins on the Arduino.

LCD → ARDUINO


AREF
GND
13
12
~11
~10
~9
8
7
~6
~5
4
~3
2
TX→1
RX←0
DIGITAL (PWM~)
L
TX
RX
ARDUINO
UNO
ON
POWER
ANALOG IN
IOREF
RESET
3.3V
5V
GND
GND
Vin
A0
A1
A2
A3
A4
A5
GND
VCC
SDA
SCL

LIQUIDCRYSTAL_I2C LIBRARY



The LCD can be programmed by using the LiquidCrystal_I2C library.

The library has functions that will make the LCD easier to control.

To install the library:

Sketch → Include Library → Manage Libraries.

This will open Arduino's Library Manager, where you can search for the library and install it.

ADDRESS

- Each I2C has an address, and this address can differ based on the manufacturer.
- The address can be found by running a program that can be found in the IDE.
- To access it, navigate to:
  - File > Examples > Wire > i2c_scanner
- The address of the I2C will be outputted to the serial monitor.
- It should look *something* like this: ***0x27***



PROGRAMMING THE LCD

```
#include <LiquidCrystal_I2C.h>  // Including the library

LiquidCrystal_I2C my_lcd = LiquidCrystal_I2C(0x27,16,2);

/*
  Above: This initializes an object named "my_lcd" using a constructor of three arguments
  Argument 1: Address
  Argument 2: Width (16 spaces)
  Argument 3: Height (2 spaces)
*/

void setup() {

  my_lcd.init();    // Initializes the LCD
  my_lcd.clear();   // Clears the LCD (empty display)
  my_lcd.backlight();   // Activates the LCD's backlight

  my_lcd.setCursor(2,0);  // Sets cursor position (x, y) (2nd space, 1st line)
  my_lcd.print("Hello world!"); // Displays "Hello world!" on the display
}
```

CUSTOM CHARACTERS

- Much like with the LED Matrix, custom characters can be made and displayed on each 5x8 sector on the LCD.
- They can be made using an array with the ***byte*** datatype and initialized by using the createChar() function.
- The createChar function takes two arguments,
  - The first argument is a number that is used to store the character. This number is then used in the write() function as an identifier.
  - The second argument is the name of the byte array that is used to create the character.

```
byte heart[8] = {
0b00000,
0b00000,
0b01010,
0b11111,
0b11111,
0b01110,
0b00100,
0b00000
};

byte key[8] = {
0b00000,
0b00110,
0b00100,
0b00110,
0b00100,
0b01110,
0b01010,
0b01110
};

LiquidCrystal_I2C my_lcd = LiquidCrystal_I2C(0x27,16,2);

void setup() {
  my_lcd.init();
  my_lcd.clear();
  my_lcd.backlight();

  my_lcd.createChar(0, key);
  my_lcd.createChar(1, heart);

  my_lcd.setCursor(7, 0);
  my_lcd.write(0);  // displays the key on the LCD

  my_lcd.setCursor(8, 0);
  my_lcd.write(1); // displays the heart on the LCD
}
```

TASK 14: MOVING CHARACTER

- Create any custom character and have it repeatedly move across the top row of the display, then the bottom row.
- Helpful resource for custom characters: https://maxpromer.github.io/LCD-Character-Creator/.
- Hint: Use a for loop.

- ***Take it further (optional):*** Alternate between the top and bottom rows so that the character looks like it's simultaneously jumping and moving across the display.

TASK 15: LIGHT INDICATOR

- Use an LDR alongside the LCD display to create a light level indicator.
- If the surroundings are bright, or if the LDR has a light shone on it, the LCD should display "Bright."
- If the surroundings are of average brightness, the LCD should display "A Little Bright."
- If the surroundings are dark, or if the LDR is covered, the LCD should display "Dark."

Bluetooth

Bluetooth and Arduino



What is Bluetooth?

- Bluetooth is a wireless technology that uses ultra-high frequency (UHS) radio frequencies that are used for communication between a master (leader) and a slave (follower) device.
- Master (leader, or central) devices search for other devices and initiate connection.
- Slave (follower, or peripheral) devices wait for connections.
- For example, when using a phone with Bluetooth-enabled headphones, the phone is the master (leader) device, and the headphones are the slave (follower) device.
- Note: The master/slave naming convention found in many technological settings is a controversial topic because they could be considered offensive. Terms like leader/follower have been implemented on other products as replacements. Be an ethical programmer 👍
- Read more here: https://www.wired.com/story/tech-confronts-use-labels-master-slave/

HC-05 Module

- The HC-05 module can be used for communication between the Arduino Uno and an android mobile device.
- The HC-05 will take the role of a follower, while your mobile device will take the role of leader. This means that the mobile device will tell the HC-05 module what to do.
- The module's name, passkey, and role can be changed in the Arduino IDE.
- It is not compatible with Apple devices.



Arduino → HC-05

VCC = 5V pin

GND = GND pin

TXD = pin 10

RXD = pin 11



When you power the Arduino, you will see a red flashing light on the bottom left corner of the front side of the HC-05 module. This indicates that the HC-05 is discoverable as a Bluetooth device and is waiting for connections.

AT Command Mode

- To enter AT command mode, disconnect power from the Arduino, then hold down the button on the bottom right of the HC-05 module as you supply power to the Arduino again.
- If the red LED on the HC-05 is blinking at 2 second intervals, the module is in AT command mode.
- This mode can be used to change the module's name, password, and role (leader, follower).
- Use this code to access the serial monitor to use AT commands.
- It checks if information from the Bluetooth module is available, and if it is, the info is printed onto the serial monitor.
- If input information from the user is available, it will be sent to the Bluetooth module.
- This allows for communication between the HC-05 and the serial monitor command line.

```
#include <SoftwareSerial.h>

SoftwareSerial BTSerial(10, 11);

void setup() {
  Serial.begin(9600);
  Serial.println("Enter AT Commands: ");
  BTSerial.begin(38400);
}

void loop() {

  if (BTSerial.available()){
    Serial.write(BTSerial.read());
  }

  if (Serial.available()){
    BTSerial.write(Serial.read());
  }

}
```

How to Use AT Command Mode

AT command mode allows a user alter the settings of the device. This happens through certain commands that are used in the serial monitor.

“

These commands use the keyword AT, then a plus sign to define a function.


File Edit Sketch Tools Help
Arduino Uno
bluetoothdemo.ino.ino
#include <SoftwareSerial.h>
SoftwareSerial BTSerial(10, 11);
void setup() {
Serial.begin(9600);
Serial.println("Enter AT Commands: ");
BTSerial.begin(38400);
}
void loop() {
if (BTSerial.available()){
Serial.write(BTSerial.read());
}
if (Serial.available()){
BTSerial.write(Serial.read());
}
}
Open Serial monitor
Both NL and CR must be selected.
The baud rate should be set to 9600
Output
Serial Monitor
AT+NAME?
Both NL & CR
9600 baud
17:47:04.735 -> Enter AT Commands:
17:54:01.403 -> OK
Serial monitor output / HC-05 module responses show up here
Serial Monitor input commands are typed here (AT commands)
Ln 10, Col 2
Arduino Uno on COM3

The Commands

Command	What it does
AT+NAME?	Returns HC-05 module's name
AT+NAME=enter name	Changes name to whatever is after the equals sign
AT+PSWD?	Returns HC-05 module's password
AT+PSWD=0000	Changes password to 0000, or whatever is after the equal's sign
AT+ROLE?	Returns the role of the HC=05 module. 0 means follower, 1 means leader, 2 means follower loop role
AT+ROLE=1	Changes the role to 1, or follower. (Should be 0 for usage)

If the command is properly executed, ***OK*** will be displayed on the serial monitor.

AT Commands – Example (Video)



MIT App Inventor

MIT Appinventor is an application that can be used to create mobile applications.

Use MIT App Inventor for Bluetooth Communication

- MIT App Inventory can be used to create mobile apps that allow for communication with the HC-05 BT module.
- Data sent from a mobile app to the HC-05 can be processed and used in the Arduino IDE to control the Arduino peripherals.
- There needs to be code in both MIT App Inventor and the Arduino IDE.
- Your device will tell the HC-05 module what to do, then the module will tell the Arduino what to do
- Code is on the next few slides.

MIT App Inventor: Getting Started

On appinventor.mit.edu, click on Create Apps, and Sign in



1. Create a new Project



2. The App Inventor Layout



- MIT App Inventor has two sections: Designer and Blocks
- Designer deals with the design/layout of the app (frontend)
- Blocks deals with the code and logic of the app (Backend)

3. Designer



- The designer interface looks like this. Elements can be dragged and dropped into the phone interface

4. Programming: Designer



- This video demonstrates how to use designer
- https://drive.google.com/file/d/171gI-UgUDnhaBo5db8iAzf0wv4C8rqA0/view?usp=sharing

5. Programming: Blocks

Video: https://drive.google.com/file/d/1H0mrlkpshmrj0xOlylKlaJe65qM7wp7g/view?usp=sharing



Set the items in the list picker to all Bluetooth devices

After the Bluetooth device is selected, connect to it

When the on button is clicked, send the byte number 49, which, in ASCII, is 1

When the off button is clicked, send the byte number 48, which, in ASCII, is 0

Change the text on the label to the connection status

6. Connect To Mobile Device

1. Ensure the HC-05 is on and blinking rapidly (connection mode)
2. Navigate to Bluetooth settings on your device, and pair with the HC-05. Use the password that you set/saw in AT command mode (usually 1234 or 0000)
3. On MIT Appinventor, on the navigation menu at the top of the screen, select `Build → Android App (apk)`
4. Scan the QR code with your device, and install the app
5. Ensure that your Bluetooth is enabled, the HC-05 has been paired, and the app has access to nearby devices (Bluetooth permissions)



MIT App Inventor – Designer and Blocks





Example: Controlling The NeoPixel ring with the HC-05 module and MIT App Inventor

```
#include <SoftwareSerial.h> // Includes the SoftwareSerial class for the code to work
#include <Adafruit_NeoPixel.h> // Includes code for Adafruit_NeoPixel class

Adafruit_NeoPixel my_ring = Adafruit_NeoPixel(16, 2, NEO_GRB + NEO_KHZ800);
// Above: Constructor for NeoPixel ring. The ring is in pin 2.


SoftwareSerial BTSerial(10, 11);  // Pin 10 is the TX pin and pin 11 is the RX pin on the HC-05 module

char incoming = 0; // Stores the value sent by MIT App Inventor

void setup() {
  Serial.begin(9600); // Sets the baud rate for both the Serial and Bluetooth device
  BTSerial.begin(9600);

  my_ring.begin(); // initializes the ring
  my_ring.clear(); // clear ring (blank)
  my_ring.show(); // send instructions to ring
}

  ------------------------------------------------------------------------------

void loop() {

  if (BTSerial.available()){  // If the Bluetooth module is available
```

```
void loop() {

  if (BTSerial.available()){  // If the Bluetooth module is available

    incoming = BTSerial.read(); // incoming is the value sent to the Bluetooth module by the app

    Serial.println(incoming); // Outputs the incoming value to the Serial monitor (not necessary)

    if (incoming == '1'){ // If incoming is 1,

      for(int i = 0; i < 16; i++){
        my_ring.setPixelColor(i,255,0,0); // sets pixel to red
        delay(50);  // 50 millisecond delay
        my_ring.show(); // updates ring, displays the light
      }

    }

    else if (incoming == '0'){  // If incoming is 0,

      for(int i = 0; i < 16; i++){
        my_ring.setPixelColor(i,0,0,0); // sets pixel to blank
        delay(50);  // 50 millisecond delay
        my_ring.show(); // updates ring, displays the light
      }
    }
  }
}
```

Congratulations!

You’re an Arduino expert!